/*
Written By: Joshua Simmerson     simmerson.4@buckeyemail.osu.edu
Date: JUN15
Summary:
Display temperature and pressure reading, while performing OSU's Phyisics 1250 heat engine lab.
*/

// Variables
int indexFIR = 0;                // Location index for FIR filters
const int num = 150;             // Number of samples for analog reads and FIR filter
float sample = 0;                // Return value of readAvg for temp and pressure readings
byte serialRead = 0;             // Input from terminal
double timer = millis();         // Timer to cycle outputs

#define tPin A1                  // Temp pin 
float temp = 999;                // Final temperature value
int tFIR[num];                   // Temp FIR filter array

#define pPin A3                  // Pressure pin
float pZoffset = 0;              // Pressure zero offest
float pressure = 999;            // Final pressure value
int pFIR [num];                  // Pressure FIR filter array

void setup()
{
  Serial.begin(9600);             // Start serial communications
  analogReference(EXTERNAL);      // Aref pin tied to 3.3V
  delay(40);                      // Wait for pressure sensor to stabilize
  sample = 0;
  for (int i = 0; i < num; i++)   // Calculate average zero offset for pressure sensor
  {
    sample += analogRead(pPin);
  }
  sample /= num;
  pZoffset = sample;               // Set offset to zero pressure reading
}

void loop()
{
 /* 
  This section is for a serial interface
   if (Serial.available())           // Check for input from terminal
   {
     serialRead = Serial.read();     // Read input
     if (serialRead==49)             // Check for flag to execute, 49 is asci for 1
     {
       calcPrint();                  // Calculate moving averages and print results to terminal
     }
   }
*/

  if (millis() > timer) {            // Print every 0.5 seconds
    calcPrint();                     // Calculate moving averages and print results to terminsl
    timer += 500;                    // Increment timer
  }
  analogRead(tPin);                       // Pre-read
  tFIR[indexFIR] = analogRead(tPin);      // Update temp FIR
  analogRead(pPin);                       // Pre-read
  pFIR[indexFIR] = analogRead(pPin);      // Update pressure FIR
  indexFIR = (++indexFIR) % num;          // Reset FIR index back to 0 when end of array is reached
}

//Average values in array to generate moving average Finite Impule Response Filter
float avgFIR(int FIR[])
{
  float avg = 0;
  for (int i = 0; i < num; i++)
  {
    avg += FIR[i];
  }
  avg /= num;
  return avg;
}

void calcPrint()
{
  // calculate pressure
  sample = avgFIR(pFIR);
  pressure = ((sample - pZoffset) * .007161458)+100;   // 100 is estimated atmospheric pressure

  // Calculate temp
  sample = avgFIR(tFIR);
  temp = .118132 * sample - 18.6736;  
  
  // Print
  Serial.print('\n');
  Serial.print("Temerature: ");
  Serial.print(temp, 1);
  //Serial.print(analogRead(tPin));
  Serial.print("C - Pressure: ");
  Serial.print(pressure);
  Serial.print("kpa");
  Serial.println('\n');

